
;***************************************************************************************************
;  溶液温度コントローラプログラム
;  tempCT
;***************************************************************************************************

LIST P=PIC16f676 ;
#INCLUDE <P16f676.INC> ;Standard Include File

__CONFIG   _CP_OFF & _CPD_OFF & _BODEN_OFF & _MCLRE_OFF & _WDT_OFF & _PWRTE_ON & _INTRC_OSC_NOCLKOUT 

;***************************************************************
;       変数の定義とレジスタ割付け
;***************************************************************

TCNT1 EQU 020H ;計測周期用カウンタ1
TCNT2 EQU 021H ;計測周期用カウンタ2
JUDG EQU 022H ;CNTLPの判定
ONDO1 EQU 023H ;温度データ上位３桁
ONDO2 EQU 024H ;温度データ下位２桁
ONDO3 EQU 025H ;温度データ下位１桁
CNT1 EQU 026H ;タイマ用カウンタ１
CNT2 EQU 027H ;タイマ用カウンタ２
CNT3 EQU 028H ;タイマ用カウンタ３
PMDAT_L EQU 029H ;ポテンショメータ値　L
PMDAT_H EQU 02AH ;ポテンショメータ値　H
SMDAT_L EQU 02BH ;サーミスタ値　L
SMDAT_H EQU 02CH ;サーミスタ値　H
SSDAT_L EQU 02DH ;温度センサ値　L
SSDAT_H EQU 02EH ;温度センサ値　H
RDBINL EQU 02FH ;読込データ値　L
RDBINH EQU 030H ;読込データ値　H

ORG 0 ;リセットベクタ（０番地）を指定

;***************************************************************
;   初期化ルーチン
;   入出力ポートの初期化
;     PORT A  RA0,1,2:アナログ入力
;             RA3:モード切替スイッチ入力
;       RA4,RC4,RC5:LEDセグメント桁制御出力
;     PORT C  RC0-RC3:LEDセグメント制御出力
;             RA5:ヒータースイッチ制御
;***************************************************************

INIT
BCF STATUS,RP0 ;バンク０に切替
CLRF PORTA ;Aポートクリア
CLRF PORTC ;Cポートクリア

MOVLW B'00000111' ;コンパレータを使わない
MOVWF CMCON

BSF STATUS,RP0 ;バンク１に切替
MOVLW B'00000111' ;RA0/AN0 RA1/AN1 RA2/AN2をアナログ入力モードに設定
MOVWF ANSEL

MOVLW B'00001111' ;RA0/AN0 RA1/AN1 RA2/AN2 RA3/AN3を入力にする
MOVWF TRISA

MOVLW B'00000000' ;Cポートをすべて出力にする
MOVWF TRISC
BCF STATUS,RP0 ;バンク０に切替

MOVLW 064H ;表示周期の初期値セット(100)
MOVWF TCNT1

;***** チャンネル切替スイッチチェック１*****

BTFSS PORTA,3 ;RA3が１なら次の命令スキップ
GOTO TEMPLP

;***************************************************************
;       室温測定処理
;***************************************************************

SENSLP
BCF PORTA,5 ;RA5を０にする(ヒーターOFF）

;****** 表示周期チェック *****

DECFSZ TCNT1,F ;ファイルレジスタTCNT1の内容をデクリメント(-1)する
GOTO DGIT1 ;結果が０でなければラベルDGIT1へ行く
MOVLW 064H ;表示周期の再セット
MOVWF TCNT1

;***** チャンネル切替スイッチチェック２ *****

BTFSS PORTA,3 ;RA3が１なら次の命令スキップ
GOTO TEMPLP ;ラベルTEMPLPへ行く
CALL GTAD2 ;センサー値読み込み
CALL CHANGE ;温度データに変換

;***** 各桁のダイナミック表示制御 *****

DGIT1
MOVF ONDO3,W ;３桁目
CALL DISP3 ;表示

MOVF ONDO2,W ;２桁目
CALL DISP2 ;表示

MOVF ONDO1,W ;１桁目
CALL DISP1 ;表示

GOTO SENSLP ;ラベルSENSLPへ行く

;***************************************************************
;       溶液温度調節処理
;***************************************************************

TEMPLP
CALL GTAD0 ;設定温度読み込み
CALL CHANGE ;温度データに変換
CLRF JUDG ;JUDGクリア

CNTLP
MOVLW 030H ;表示周期の初期値セット
MOVWF TCNT2

;***** 各桁のダイナミック表示制御 ******

DGIT2
MOVF ONDO3,W ;３桁目
CALL DISP3 ;表示

MOVF ONDO2,W ;２桁目
CALL DISP2 ;表示

MOVF ONDO1,W ;１桁目
CALL DISP1 ;表示

DECFSZ TCNT2,F ;ファイルレジスタTCNT2の内容をデクリメント(-1)する
GOTO DGIT2 ;結果が０でなければラベルDGIT2へ行く
DECF JUDG,F ;JUDG-1
BTFSC STATUS,Z ;結果が0なら次の命令スキップ
GOTO CHECK1

;***** チャンネル切替スイッチチェック３ *****

CALL TIM50M
BTFSC PORTA,3 ;RA3が０なら次の命令スキップ
GOTO SENSLP

;********************************************

CALL GTAD1 ;サーミスタ値読み込み
CALL CHANGE ;温度データに変換
MOVLW 01H
MOVWF JUDG ;JUDGに１をセット
GOTO CNTLP

CHECK1
MOVF SMDAT_L,W
SUBWF PMDAT_L,F ;PMDATA_L-SMDAT_L
BTFSC STATUS,C ;Cビットが０ならスキップ
GOTO CHECK2 ;結果が正ならCHECK2へ
MOVLW 01H
SUBWF PMDAT_H,F ;PMDAT_H-1
BTFSS STATUS,C ;Cビットが１ならスキップ
GOTO CHECK3 ;結果が０ならCHECK3へ

CHECK2
MOVF SMDAT_H,W
SUBWF PMDAT_H,F ;PMDAT_H-SMDAT_H
BTFSS STATUS,C ;Cビットが１ならスキップ
GOTO CHECK3 ;結果が０ならCHECK3へ
BSF PORTA,5 ;RA5を１にする(ヒーターON）
CALL TIM500M
GOTO TEMPLP

CHECK3
BCF PORTA,5 ;RA5を０にする(ヒーターOFF）
CALL TIM500M
GOTO TEMPLP

;***************************************************************
;******************   サブルーチン群   *************************
;***************************************************************

;***************************************************************
;      データ変換サブルーチン
;      ２バイトバイナリデータを３桁温度データへ
;***************************************************************

CHANGE
CLRF ONDO1 ;１桁目クリア
CLRF ONDO2 ;２桁目クリア
CLRF ONDO3 ;３桁目クリア

;****** 100位を求める *****

BD100
MOVLW 064H ;100
SUBWF RDBINL,W ;RDBIN_L-100
BTFSS STATUS,C ;Cビットが１ならスキップ
GOTO BD101 ;結果が０ならBD101へ
MOVWF RDBINL ;LOWDATA更新
INCF ONDO3,F ;100位+1
GOTO BD100

;***** 上位バイトから引く ******

BD101
MOVF RDBINH,F ;上位バイト0か？
BTFSC STATUS,Z ;Zビットが０ならスキップ
GOTO BD010 ;すでに0で10位に進む
MOVWF RDBINL ;LOWDATA更新
INCF ONDO3,F ;まだ引ける
DECF RDBINH,F ;UPDATA-1
GOTO BD100

;***** 10位を求める *****

BD010
MOVLW 0AH ;10
SUBWF RDBINL,W ;LOWDATA-10
BTFSS STATUS,C ;Cビットが１ならスキップ
GOTO BD001 ;結果が０ならBD101へ
MOVWF RDBINL ;LOWDATA更新
INCF ONDO2,F ;10位+1
GOTO BD010

;***** 1位を求める *****

BD001
MOVF RDBINL,W
MOVWF ONDO1
RETURN

;***************************************************************
;       CH0　A/D変換、設定温度(PM値)読み込み
;***************************************************************

GTAD0
MOVLW B'00000001' ;A/D電源ON(左桁揃え AN0)
MOVWF ADCON0
CALL TIM20
BSF ADCON0,1 ;A/D変換を開始

WAIT1
BTFSC ADCON0,1 ;A/D変換が終了するのを待つ
GOTO WAIT1
MOVF ADRESH,W ;A/D変換結果をPMDAT_Hに書く
MOVWF PMDAT_H
BSF STATUS,RP0 ;バンク１に切替
MOVF ADRESL,W ;A/D変換結果をPMDAT_Lに書く
BCF STATUS,RP0 ;バンク０に切替
MOVWF PMDAT_L
BCF STATUS,C ;Cフラグクリア
RLF PMDAT_L,W ;下位データ最上位桁をCフラグへ（表示は9BIT）
RLF PMDAT_H,W ;Cフラグを上位桁最下位ビットへ
MOVWF PMDAT_L ;W→PMDAT_L
MOVWF RDBINL ;W→RDBINL
CLRF RDBINH ;RDBINHをクリア
RLF RDBINH,W ;Cフラグを上位へ
MOVWF RDBINH ;W→RDBINH
MOVWF PMDAT_H ;W→PMDAT_H
BCF ADCON0,0 ;A/D変換を終了
RETURN

;***************************************************************
;       CH1　A/D変換、サーミスタ値読み込み

;***************************************************************

GTAD1
MOVLW B'10000101' ;A/D電源ON(右桁揃え AN0)
MOVWF ADCON0
CALL TIM20
BSF ADCON0,1 ;A/D変換を開始

WAIT2
BTFSC ADCON0,1 ;A/D変換が終了するのを待つ
GOTO WAIT2
MOVF ADRESH,W ;A/D変換結果をSMDAT_Hに書く
MOVWF SMDAT_H
BSF STATUS,RP0 ;バンク１に切替
MOVF ADRESL,W ;A/D変換結果をSMDAT_Lに書く
BCF STATUS,RP0 ;バンク０に切替
MOVWF SMDAT_L
MOVLW 0C8H ;補正値を入力
SUBWF SMDAT_L,W ;SMDAT_L-0C8H(200）
MOVWF SMDAT_L
BTFSS STATUS,C ;Cビットが１ならスキップ
GOTO JUMP1 ;減算の結果が負ならJUMP1
MOVF SMDAT_H,W ;SMDAT_H→W
MOVWF RDBINH ;W→RDBINH
MOVF SMDAT_L,W ;SMDAT_L→W
GOTO JUMP2 ;JUMP2へ

JUMP1
DECF SMDAT_H,W ;SMDAT_H-1
MOVWF SMDAT_H ;W→SMDAT_H
MOVWF RDBINH ;W→RDBINH
MOVF SMDAT_L,W ;SMDAT_L→W

JUMP2
MOVWF RDBINL ;W→RDBINL
BCF ADCON0,0 ;A/D変換を終了
RETURN

;***************************************************************
;       CH2　A/D変換、センサー(LM35D)値読み込み
;***************************************************************

GTAD2
MOVLW B'00001001' ;A/D電源ON(左桁揃え AN0)
MOVWF ADCON0
CALL TIM20
BSF ADCON0,1 ;A/D変換を開始

WAIT3
BTFSC ADCON0,1 ;A/D変換が終了するのを待つ
GOTO WAIT3
MOVF ADRESH,W ;A/D変換結果をSSDAT_Hに書く
MOVWF SSDAT_H
BSF STATUS,RP0 ;バンク１に切替
MOVF ADRESL,W ;A/D変換結果をSSDAT_Lに書く
BCF STATUS,RP0 ;バンク０に切替
MOVWF SSDAT_L
BCF STATUS,C ;Cフラグクリア
RLF SSDAT_L,W ;下位データ最上位桁をCフラグへ（表示は9BIT）
RLF SSDAT_H,W ;Cフラグを上位桁最下位ビットへ
MOVWF RDBINL ;W→RDBINL
CLRF RDBINH ;RDBINHをクリア
RLF RDBINH,F ;Cフラグを上位へ
BCF ADCON0,0 ;A/D変換を終了
RETURN

;***************************************************************
;       セグメントLED表示制御サブルーチン
;       約５msec周期のダイナミック点灯制御
;***************************************************************

DISP3
MOVWF PORTC ;セグメント出力
BSF PORTA,4 ;RA4を１にする
CALL TIM5M ;Wait 5msec
CLRF PORTC ;セグメント消去
BCF PORTA,4 ;RA4を０にする
RETURN

DISP2
MOVWF PORTC ;セグメント出力
IORLW 020H ;Wレジスタの内容 OR 20H
MOVWF PORTC ;桁制御出力
CALL TIM5M ;Wait 5msec
CLRF PORTC ;セグメント消去
RETURN

DISP1
MOVWF PORTC ;セグメント出力
IORLW 010H ;Wレジスタの内容 OR 10H
MOVWF PORTC ;桁制御出力
CALL TIM5M ;Wait 5msec

CLRF PORTC ;セグメント消去
RETURN

;***************************************************************
;       タイマールーチン
;***************************************************************

;*****  Timer 20us  *****

TIM20 ; 2+16+2 = 20cycle
MOVLW 05H ;1cycle
MOVWF CNT1 ;1cycle

TIMLP1 ; 3*5+1 = 16cycle
DECFSZ CNT1,F ;1(no skip), 2
GOTO TIMLP1 ;2cycle
RETURN ;2cycle

;*****  Timer 5ms  *****

TIM5M ; 2+4999+2 = 5003cycle
MOVLW 05H ;1cycle
MOVWF CNT1 ;1cycle

TIMLP2 ; (995+5)*5+1-2 = 4999cycle
MOVLW 0F9H ;1cycle
MOVWF CNT2 ;1cycle

TIMLP3 ; 4*249+1-2 = 995cycle
NOP ;1cycle
DECFSZ CNT2,F ;1(no skip), 2
GOTO TIMLP3 ;2cycle
DECFSZ CNT1,F ;1(no skip), 2
GOTO TIMLP2 ;2cycle
RETURN ;2cycle

;*****  Timer 50ms  *****

TIM50M ; 2+49999+2 = 50003cycle
MOVLW 032H ;1cycle
MOVWF CNT1 ;1cycle

TIMLP4 ; (995+5)*50+1-2 = 49999cycle
MOVLW 0F9H ;1cycle
MOVWF CNT2 ;1cycle

TIMLP5 ; 4*249+1-2 = 995cycle
NOP ;1cycle
DECFSZ CNT2,F ;1(no skip), 2
GOTO TIMLP5 ;2cycle
DECFSZ CNT1,F ;1(no skip), 2
GOTO TIMLP4 ;2cycle
RETURN ;2cycle

;*****  Timer 500ms  *****

TIM500M ; 2+500079+2 = 500083cycle
MOVLW 0AH ;1cycle
MOVWF CNT3 ;1cycle

TIMLP6 ; (50003+5)*10+1-2 = 500079cycle
CALL TIM50M ;2cycle
DECFSZ CNT3,F ;1(no skip), 2
GOTO TIMLP6 ;2cycle
RETURN ;2cycle

END

- 5 -


